Analisi spettrale di segnali deterministici

Marcello Colozzo

Esercizio 1

Determinare la densità spettrale e lo spettro di potenza del segnale $V(t) = V_0$ se $|t| \leq \tau$, 0 se $|t| > \tau$

Calcoli manuali

```
SetOptions[
  {
   Plot,
   ListLinePlot
  },
  TicksStyle -> Directive[
    Hue[5 / 6],
    8
   ]
 ];
```

```
V[t_, V0_, τ_] := Which[
  t ≥ -τ && t ≤ τ, V0,
  t > τ || t < -τ, 0
 ]
```

```
segnale[V0_, τ_] := Plot[
  V[t, V0, τ],
  {t, -4 τ, 4 τ},
  Exclusions → {t == τ, t == -τ},
  PlotRange → {-0.1, V0 + 0.8},
  Ticks →
   {
    {
     {-τ, "-τ"}, {τ, "τ"}
    },
    {
     {V0, "V0"}
    }
   },
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["t", Small, Red],
    Style["V", Small, Blue]
   },
  Epilog → {
    Red,
    Dashed,
    Line[{{-τ, 0}, {-τ, V0}}],
    Line[{{τ, 0}, {τ, V0}}]
   }
 ]
```

```
segnale[1, 2]
```



La trasformata di Fourier è $\hat{V}(\omega) = \mathrm{V0}/\ \mathrm{Sqrt}[2\,\pi] \int_{-\tau}^{+\tau} \mathrm{Exp}(-j\,\omega\,t)\,dt = \frac{\mathrm{V0}}{\sqrt{2\,\pi}} * \frac{\sin(\omega\,\tau)}{\omega}$, per cui la densità spettrale è

$$\rho[\omega\_,\ \mathrm{V0}\_,\ \tau\_] := \frac{2\ \mathrm{V0}}{\sqrt{2\,\pi}} * \frac{\mathrm{Sin}[\omega\,\tau]}{\omega}$$

```
densitaspettrale[V0_, τ_] := Plot[
  ρ[ω, V0, τ],
  {ω, -10/τ, 10/τ},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["ρ", Small, Blue]
   }

 ]
```

```
densitaspettrale[1, 2]
```



Lo spettro di potenza è:

```
w[ω_, V0_, τ_] := (4 V0^2)/(2 π) * ((Sin[ω τ])^2)/(ω^2)
```

```
spettropotenza[V0_, τ_] := Plot[
  w[ω, V0, τ],
  {ω, -10/τ, 10/τ},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["w", Small, Blue]
   }
 ]
```

```
spettropotenza[1, 2]
```



Calcoli automatici

Per il calcolo della trasformata di Fourier possiamo utilizzare il comando **FourierTransform**

```
?FourierTransform
```

FourierTransform[*expr*, *t*, *ω*] gives the symbolic Fourier transform of *expr*.
FourierTransform[*expr*, {t_1, t_2, …}, {ω_1, ω_2, …}] gives the multidimensional Fourier transform of *expr*. ≫

I parametri (a, b) sono così definiti: $\sqrt{\frac{|b|}{(2\pi)^{1-a}}}\int_{-\infty}^{+\infty} V(t)\, e^{j\,b\,\omega t}\, dt$, per cui nel nostro caso è $a = 0$, $b = -1$

```
ρ1[ω_, V0_, τ_](*definizione immediata per
   non aumentare il carico computazionale*) = FourierTransform[
   (*funzione da trasformare*)
   V[t, V0, τ],
   (*variabile indipendente*)
   t,
   (*variabile indipendente della trasformata*)
   ω,
   (*settaggio dei parametri*)
   FourierParameters → {0, -1}
  ];
```

```
ρ1[ω, V0, τ]
```

$$-\frac{\sqrt{\frac{2}{\pi}}\ \mathrm{V0}\ \mathrm{Sin}[\tau\,\omega]\ (-1 + \mathrm{UnitStep}[-\tau])}{\omega}$$

```
Plot[
 ρ1[ω, 2, 1],
 {ω, -10, 10}
]
```



Se avessimo usato la definizione ritardata:

```
Clear[ρ1]
```

```
ρ1[ω_, V0_, τ_] := FourierTransform[
  (*funzione da trasformare*)
  V[t, V0, τ],
  (*variabile indipendente*)
  t,
  (*variabile indipendente della trasformata*)
  ω,
  (*settaggio dei parametri*)
  FourierParameters → {0, -1}
 ]
```

Impiega troppo tempo se tracciamo il grafico:

```
Plot[
  ρ1[ω, 2, 1],
  {ω, -10, 10}
 ] // Timing
```

```
$Aborted
```

Quando l'espressione analitica della funzione da trasformare è troppo complicata (non è questo il caso), è preferibile calcolare la trasformata per via numerica. A tale scopo carichiamo il package `FourierSeries:`

```
Needs["FourierSeries`"]
```

```
NFourierTransform[
 (*funzione da trasformare*)
 V[t, 1, 2],
 (*variabile indipendente della funzione*)
 t,
 (*valore numerico della variabile indipendente della trasformata*)
 1,
 (*settaggio dei parametri*)
 FourierParameters → {0, -1}
]
```

$0.725514 - 5.53644 \times 10^{-18}\, i$

```
Plot[
 NFourierTransform[
  V[t, 1, 2],
  t,
  ω
 ],
 {ω, -4, 4}
]
```




```
Clear[V, w, ρ, ρ1, segnale, spettropotenza, densitaspettrale]
```

Esercizio 2

Determinare la densità spettrale e lo spettro di potenza del segnale $V(t) = V_0\, e^{-\omega_0 |t|}$ con $\omega_0 > 0$.

Calcoli manuali

```
V[t_, V0_, ω0_] := V0 * ⅇ^(-ω0*Abs[t])
```

```
segnale[V0_, ω0_] := Plot[
  V[t, V0, ω0],
  {t, -4/ω0, 4/ω0},
  PlotRange → {-0.1, V0 + 0.4},
  PlotStyle → {Thickness[0.003]},
  AxesLabel →
   {
    Style["t", Small, Red],
    Style["V", Small, Blue]
   },
  Ticks →
   {
    None,
    {
     {V0, "V0"}
    }
   }
 ]
```

```
segnale[1, 2]
```



In questo caso l'integrale di Fourier si spezza nella somma di due integrali, poiché conviene esplicitare il valore assoluto. Lo svolgimento dei calcoli è raccolto in questo link.

Verifichiamo:

```
I1[ω_] = Integrate[
  Exp[(ω0 - ⅈ ω) * t],
  {t, -∞, 0},
  Assumptions →
   {
    ω ∈ Reals,
    ω0 > 0
   }
 ]
```

$$\frac{1}{-\mathrm{i}\,\omega + \omega 0}$$

```
I2[ω_] = Integrate[
  Exp[-(ω0 + ⅈ ω) * t],
  {t, 0, +∞},
  Assumptions →
   {
    ω ∈ Reals,
    ω0 > 0
   }
 ]
```

$$\frac{1}{\mathrm{i}\,\omega + \omega 0}$$

La densità spettrale è:

```
ρ[ω_, V0_, ω0_] = V0/Sqrt[2 π] * (I1[ω] + I2[ω]) // Simplify
```

$$\frac{\sqrt{\frac{2}{\pi}}\ \mathrm{V0}\ \omega 0}{\omega^2 + \omega 0^2}$$

Lo spettro di potenza

```
w[ω_, V0_, ω0_] = ρ[ω, V0, ω0]^2
```

$$\frac{2\ \mathrm{V0}^2\ \omega 0^2}{\pi\ \left(\omega^2 + \omega 0^2\right)^2}$$

```
Clear[densitaspettrale]
```

```
densitaspettrale[V0_, ω0_] := Plot[
  ρ[ω, V0, ω0],
  {ω, -4 ω0, 4 ω0},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["ρ", Small, Blue]
   }

 ]
```

```
densitaspettrale[1, 2]
```




```
spettropotenza[V0_, ω0_] := Plot[
  w[ω, V0, ω0],
  {ω, -4 ω0, 4 ω0},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["w", Small, Blue]
   }
 ]
```

```
spettropotenza[1, 2]
```



Calcoli automatici

```
ρ1[ω_, V0_, ω0_](*definizione immediata per
   non aumentare il carico computazionale*) = FourierTransform[
   (*funzione da trasformare*)
   V[t, V0, ω0],
   (*variabile indipendente*)
   t,
   (*variabile indipendente della trasformata*)
   ω,
   (*settaggio dei parametri*)
   FourierParameters → {0, -1}
  ];
```

```
ρ1[ω, V0, ω0]
```

$$\frac{\sqrt{\frac{2}{\pi}}\, V0\, \omega 0}{\omega^2 + \omega 0^2}$$

```
ρ1[ω, 2, 1]
```

$$\frac{2\sqrt{\frac{2}{\pi}}}{1 + \omega^2}$$

```
Plot[
 ρ1[ω, 2, 1],
 {ω, -10, 10},
 PlotRange → All
]
```




```
Clear[V, w, ρ, ρ1, segnale, spettropotenza, densitaspettrale]
```

Esercizio 3

Determinare la densità spettrale e lo spettro di potenza del segnale $V(t) = \frac{A}{\tau}$ se $\left|t\right| \leq \frac{\tau}{2}$, 0 se $\left|t\right| > \frac{\tau}{2}$. Discutere il limite per $\tau \to 0^{+}$.

Calcoli manuali

```
V[t_, A_, τ_] := Which[
   t ≥ -τ/2 && t ≤ τ/2, A/τ,
   t > τ/2 || t < -τ/2, 0
  ]
```

```
segnale[A_, τ_] := Plot[
  V[t, A, τ],
  {t, -2 τ, 2 τ},
  Exclusions → {t == τ/2, t == -τ/2},
  PlotRange → {-0.1, A/τ + 1.8},
  Ticks →
   {
    {
     {-τ/2, "-τ/2"}, {τ, "τ/2"}
    },
    {
     {A/τ, "A/τ"}
    }
   },
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["t", Small, Red],
    Style["Vτ", Small, Blue]
   },
  Epilog → {
    Red,
    Dashed,
    Line[{{-τ/2, 0}, {-τ/2, A/τ}}],
    Line[{{τ/2, 0}, {τ/2, A/τ}}]
   }
 ]
```

```
segnale[1, 0.1]
```



I calcoli manuali della trasformata di Fourier sono raccolti in questo file.

Verifichiamo:

```
ρ[ω_, A_, τ_] = 1/Sqrt[2 π] * A/τ Integrate[
   Exp[-ⅈ ω * t],
   {t, -τ / 2, +τ / 2},
   Assumptions →
    {
     ω ∈ Reals,
     A > 0,
     τ > 0
    }
  ]
```

$$\frac{A \sqrt{\frac{2}{\pi}} \operatorname{Sin}\left[\frac{\tau\,\omega}{2}\right]}{\tau\,\omega}$$

Lo spettro di potenza è

```
w[ω_, A_, τ_] = ρ[ω, A, τ]^2
```

$$\frac{2 A^2 \operatorname{Sin}\left[\frac{\tau\,\omega}{2}\right]^2}{\pi\,\tau^2\,\omega^2}$$

```
densitaspettrale[A_, τ_] := Plot[
  ρ[ω, A, τ],
  {ω, -100 τ, 100 τ},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["ρτ", Small, Blue]
   },
  PlotRange → All

 ]
```

```
densitaspettrale[1, 0.5]
```




```
spettropotenza[A_, τ_] := Plot[
  W[ω, A, τ],
  {ω, -100 τ, 100 τ},
  PlotStyle → {
    Thickness[0.003]
   },
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["W", Small, Blue]
   },
  PlotRange → All
 ]
```

```
spettropotenza[1, 0.5]
```



Eseguiamo il limite per $\tau \to 0^+$ (segnale impulso unitario) :

```
Limit[
 ρ[ω, A, τ],
 τ → 0
]
```

$$\frac{A}{\sqrt{2\pi}}$$

```
Clear[V, w, ρ, ρ1, segnale, spettropotenza, densitaspettrale]
```

Esercizio 4

Determinare la densità spettrale e lo spettro di potenza del segnale modulato in ampiezza $V(t) = V_0\, e^{-\Omega^2{}_0\, t^2} \cos(\omega_0\, t)$

```
A[t_, Ω_, V0_] := V0 * Exp[-Ω^2 * t^2]
```

```
V[t_, V0_, Ω_, ω0_] := A[t, Ω, V0] * Cos[ω0 * t]
```

```
segnale = Plot[
  {
   V[t, 1, 10, 150],
   A[t, 10, 1],
   -A[t, 10, 1]
  },
  {t, -0.5, 0.5},
  PlotRange → All,
  PlotStyle →
   {
    Thickness[0.003],
    {
     RGBColor[0.3, 0.6, 0.7],
     Thickness[0.001]
    },
    {
     RGBColor[0.3, 0.6, 0.7],
     Thickness[0.001]
    }
   },
  Ticks →
   {
    None,
    {
     {-1, "-V0"}, {1, "V0"}
    }
   },
  AxesLabel →
   {
    Style["t", Small, Red],
    Style["V", Small, Blue]
   }
 ]
```



I calcoli dell'integrale di Fourier sono molto laboriosi ed occorre l'applicazione del teorema di Cauchy per una funzione olomorfa definita nel dominio in figura. Per i dettagli consultare il file al seguente link.

```
dominio = Plot[
  Null,
  {x, -2, 2},
  AxesLabel →
   {
    Style["x", Small, Red],
    Style["y", Small, Blue]
   },
  Ticks → {
    {
     {-1.5, "-x0"}, {1.5, "x0"}
    },
    {
     {0.25, "y0"}
    }
   },
  Epilog →
   {
    RGBColor[0.3, 0.6, 0.7],
    Thickness[0.004],
    Arrow[{{-1.5, 0}, {1.5, 0}}],
    Arrow[{{1.5, 0}, {1.5, 0.25}}],
    Arrow[{{1.5, 0.25}, {-1.5, 0.25}}],
    Arrow[{{-1.5, 0.25}, {-1.5, 0}}],
    Text["D", {0.2, 0.1}]
   }

 ]
```



La densità spettrale è:

```
ρ[ω_, V0_, Ω_, ω0_] := V0/(Ω * √2) Exp[-(ω - ω0)^2/(4 Ω^2)]
```

```
densitaspettrale = Plot[
  ρ[ω, 1, 10, 150],
  {ω, 100, 200},
  PlotRange → All,
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["ρ", Small, Blue]
   },
  Ticks → {
    {
     {150, "ω0"}
    },
    {
     {1/(10 Sqrt[2]), "V0/(Ω Sqrt[2])"}
    }
   },
  PlotStyle → Thickness[0.003]

 ]
```



Verifichiamo:

```
V[t, V0, Ω, ω0]
```

$e^{-t^2 \Omega^2}$ V0 Cos[t ω0]

```
ρ1[ω_, V0_, Ω_, ω0_](*definizione immediata per
   non aumentare il carico computazionale*) = FourierTransform[
   (*funzione da trasformare*)
   V[t, V0, Ω, ω0],
   (*variabile indipendente*)
   t,
   (*variabile indipendente della trasformata*)
   ω,
   (*settaggio dei parametri*)
   FourierParameters → {0, -1}
  ];
```

```
ρ1[ω, V0, Ω, ω0]
```

$$V0\left(\frac{\text{Cosh}\left[\frac{(\omega-\omega 0)^2}{4\,\Omega^2}\right]}{2\sqrt{2}\,\text{Abs}[\Omega]}+\frac{\text{Cosh}\left[\frac{(\omega+\omega 0)^2}{4\,\Omega^2}\right]}{2\sqrt{2}\,\text{Abs}[\Omega]}-\frac{\text{Sinh}\left[\frac{(\omega-\omega 0)^2}{4\,\Omega^2}\right]}{2\sqrt{2}\,\text{Abs}[\Omega]}-\frac{\text{Sinh}\left[\frac{(\omega+\omega 0)^2}{4\,\Omega^2}\right]}{2\sqrt{2}\,\text{Abs}[\Omega]}\right)$$

Un risultato strano!

```
Plot[
 ρ1[ω, 1, 10, 150],
 {ω, 100, 200},
 PlotRange → All
]
```



Proviamo a calcolare l'integrale:

```
J[ω_, V0_, Ω_, ω0_] = Integrate[
  Exp[-Ω^2 t^2] * Exp[ⅈ (ω0 - ω) * t],
  {t, -∞, +∞},
  Assumptions →
   {
    ω ∈ Reals,
    ω0 > 0,
    Ω > 0
   }
 ]
```

$$\frac{e^{-\frac{(\omega-\omega 0)^2}{4\,\Omega^2}}\sqrt{\pi}}{\Omega}$$

quindi abbiamo svolto bene i conti. In particolare l'integrale il cui calcolo ha richiesto l'applicazione del teorema di Cauchy:

```
I1[ω_, V0_, Ω_, ω0_] = Integrate[
  Exp[-Ω^2 (ⅈ * (ω0 - ω)/(2 Ω^2) - t)^2],
  {t, -∞, +∞},
  Assumptions →
   {
    ω ∈ Reals,
    ω0 > 0,
    Ω > 0
   }
 ]
```

$$\frac{\sqrt{\pi}}{\Omega}$$

```
Clear[segnale, spettropotenza, densitaspettrale, plotN]
```

Per $\omega_0 = 0$

```
segnale = Plot[
  {
   V[t, 1, 10, 0]
  },
  {t, -0.5, 0.5},
  PlotRange → All,
  PlotStyle →
   {
    Thickness[0.003]
   },
  Ticks →
   {
    None,
    {
     {-1, "-V0"}, {1, "V0"}
    }
   },
  AxesLabel →
   {
    Style["t", Small, Red],
    Style["V", Small, Blue]
   }
 ]
```




```
Export["gaussiano.eps", segnale]
```

```
gaussiano.eps
```

La trasformata di Fourier è ancora una gaussiana, ma con larghezza inversamente proporazionale alla larghezza del segnale:

```
densitaspettralegaussiana = Plot[
  ρ[ω, 1, 10, 0],
  {ω, -50, 50},
  PlotRange → All,
  AxesLabel →
   {
    Style["ω", Small, Red],
    Style["ρ", Small, Blue]
   },
  Ticks → {
    {
     {150, "ω0"}
    },
    {
     {1/(10 √2), "V0/(Ω √2)"}
    }
   },
  PlotStyle → Thickness[0.003]

 ]
```




```
Clear[V, w, ρ, ρ1, segnale, spettropotenza, densitaspettrale, plotN]
```

Trasformata discreta di Fourier

L'analisi spettrale di un segnale deterministico può essere eseguita utilizzando la trasformata discreta di Fourier.

```
? Fourier
```

Fourier[*list*] finds the discrete Fourier transform of a list of complex numbers. ≫

Ad esempio:

```
V[t_, V0_, Ω_] := V0 * Exp[-Ω * Abs[t]^3] / (1 + Ω Abs[t])
```

```
segnale = Plot[
  V[t, 1, 10],
  {t, -2, 2},
  PlotRange → All
 ]
```



Eseguendo un campionamento discreto:

```
Vd[k_, V0_, Ω_] := V0 * Exp[-Ω * Abs[k]^5] / (1 + Ω Abs[k])
```

```
Vd[0, 1, 10]
```

```
1
```

```
segnalediscreto = Table[
   {k, Vd[k, 1, 10]},
   {k, -2, 2, 10^-1}
  ];
```

```
segnalediscretoplot = ListLinePlot[
  segnalediscreto,
  PlotRange → All
 ]
```



La trasformata di Fourier discreta:

```
DFT[n_] := Fourier[
   Table[
    Vd[k, 1, 10],
    {k, -n, n}
   ]
  ];
```

Per plottare la densità spettrale costruiamo un insieme di coppie ordinate $(k, \hat{V})$:

```
lista[n_, k_] := {k, Abs[DFT[n][[k]]]}
```

```
lista1[n_] := Table[
  lista[n, k],
  {k, -n, n}
 ]
```

```
densitaspettraleplot = ListLinePlot[
  lista1[40]
 ]
```